
ABSTRACT 
PROC BOXPLOT is a good and efficient method for exploring data distributions (the mean, quartiles, minimum 
and maximum observations for a group), but the style is limited to skeletal, schematic and schematicfar. If you 
want to show the inference (Mean±SE/Mean±SD) as the upper/lower whisker as well as specific patient 
information or the user prepared percentile with upper/lower whisker, you can not count on PROC BOXPLOT 
to produce this kind of plot, and this is when PROC GPLOT and the annotate facility come in handy.  This paper 
will show how to produce customized box plots and is targeted to the intermediate level audience. 

INTRODUCTION 
The annotate facility is a useful tool for generating customized plots. One typical use is to label the treatment 
group by the survival curve or any other line plot. Another use is to replace the legend. The use of the annotate 
facility is through the annotate dataset.  The value of variables in each observation provides instruction on how 
to construct the graph such as plotting the symbol, labeling the data points, drawing the lines, etc. Some 
important variables are FUNCTION, XSYS, YSYS, X, Y, TEXT, POSITION, STYLE, etc. These variable 
values will determine what, where, and when to draw. By constructing a carefully designed annotate dataset, we 
can produce a tailored box plot according to the client’s specific need.  
 
BACKGROUND: 
 In order to understand the data, statisticians often want to combine descriptive statistics (mean, median, 25 
percentile, 75 percentile) and inferential statistics (standard error of the mean or standard deviation) into one 
boxplot. Also, they may want to see some special patient’s data on the boxplot. Conventional PROC BOXPLOT 
cannot accomplish this so we need the annotate facility combined with PROC GPLOT to achieve this 
customization. 
 
For example, in one of our studies, most data of one primary endpoint are on 24 hour duration, but only a few 
patients' data are calculated on the very short duration.  The statistician would like to have a boxplot to show 
these special patients' information as well as the standard deviation. In all of these cases, we have to use the 
annotate facility to customize the boxplot. In the following sections, we will provide three examples to illustrate 
a step-by-step guide to build your own box plot. 
 
DATA SETUP 
First, we need to set up the analysis data named as tstv. In our example, we are setting up five escalating dose 
treatment groups, 10 mg, 20 mg, 30 mg, 100 mg and 200 mg. There are different numbers of subjects in each 
treatment group.  
The input tstv dataset contains the following information: 
 

Treatment Group User ID Exam value 
Group Usubjid tstv 
1 1001 100.28530882 
… … … 
2 2001 76.788510135 

 
THE METHOD AND PROCEDURES 
Next, we will look at several examples of customized box plots. 
 



Example 1 
Use proc boxplot procedure: the style is limited to skeletal, schematic and schematicfar. The skeletal is 
the default style in which the upper/lower whisker is the maximum and  minimum respectively.  In the 
schematic style, the upper whisker represents the observed maximum value less than Q3(quartile 
3)+1.5*IQR(interquartile range), the lower whisker represent the minimum value great than Q1(quartile 
1)-1.5*IQR. The following skeletal type of boxplot shows min, max, median, mean, Q1, Q3 summary 
statistics. 

axis1 order = (1 t 5 by 1) value=(f=swissb h=0.8) offset=(2,1) o  
label = (height = 1.25 'Treatment')minor = none; 
axis2 order = (60 to 130 by 10) value=(f=swissb h=0.8) 
offset=(2,2)label = (height = 1.25 Endpoint') ; '
symbol value = dot cv=red height = 1 ; 
 

proc boxplot data 
= tstv; 
PLOT tstv*group / 
boxstyle=skeletal 
haxis=axis1 
vaxis=axis2 
cboxfill=TAN 
cboxes = BL  
; 
title h=1 
"upper/lower 
whisker represent 
Max/Min Value"; 
insetgroup mean 
(5.1) min max / 
header = 'Stats by 
Treatment' 
pos = top 
cfi ll  = YELLOW ; 
run; 

Example 2: 

Produce side-by-side boxplots by using the basic commands in PROC GPLOT with options selected 
for the SYMBOL statement: INTERPOL= BOXT10 specifies a box plot with tops and bottoms on its 
whiskers (T) and the 10 implies high and low bounds at the 90th and 10th percentiles.  The following 
plot displays the min, max, median, Q1, Q3, 10thand 90th percentile summary statistics along with the 
specific patient's data point included in annotate dataset: 
 

1. Create the annotate dataset called "special" used for plotting the specific patient's information.  
Dataset "special" contains the select specific patient/s information (in this example, patient 3007)  
%let extend=0.07; /*choose a value to extend from middle point to  
                    draw the box */ 

group Usubjid tstv function color style Text xsys ysys size position 
3 3007 113.6 SYMBOL black MARKER V 2 2 1 5 
3 3007 113.6 LABEL black CENTB AN=3007 2 2 1 3 
3 3007 113.6 SYMBOL black MARKER V 2 2 1 3 

 
… usubjid x y Description for the variables: position, x, y 
… 3007 3 113.6 x=group ;  y=tstv ; position=5:center ; xsys=ysys=2: actually value 
… 3007 3.12 113.6 x=group+&extend+0.05 ;  y=tstv ; position=3: top 
… 3007 3.09 114.6 x=group+&extend+0.02; ; y=tstv+1 

 



2. Output plot using proc gplot and annotate dataset.  
      SYMBOL interpol=boxt10 bwidth=3 width=1 color=blue 

       value=dot cv=red height=1; 
      axis1 order = (1 to 5 by 1) offset=(3,3) 
            label = ( ight 1.25 Treatment'  inor=none; he  =  ' ) m

axis2 order = (60 to 130 by 10) offset=(3,3) 
      label = (height = 1.25 'Endpoint') ; 

PROC GPLOT 
DATA=tstv; 
PLOT tstv*group/ 
anno=special 
haxis=axis1 
vaxis=axis2;  
FORMAT group Grp.; 
TITLE H=1.5 " 
upper/lower 
whisker represent 
the 90th & 10th 
percentiles"; 
RUN; QUIT; 

 
Example 3 
Customize the boxplot to include mean, median, Q1, Q3, inferential statistics (Mean+SD as upper bound and 
Mean-SD as low bound), outliers, and special patients' information of user defined criteria. 

 
• Calculate the sample size for each dose treatment group and store that resulting calculation in the macro 

variables: cn1-cn5. 
• Capture summary statistics, the necessary percentiles and choose your own upper and lower whisker. In our 

example, we have chosen Mean±SD as the upper and lower whisker. You can choose your own bound or 
even write a macro to provide flexibility. 

 
proc univariate data=tstv noprint; 
 by group; 
 var tstv; 
 output out=sumtstv1 
        std=std q1=q1 median=median mean=mean q3=q3 min=min max=max 
   p95=p95 p5=p5 p99=p99 p1=p1 p90=p90 p10=p10; 
run; 
data sumtstv; 
 set sumtstv1; 
 upper=mean+std; 
 lower=mean-std; 
run; 
proc sort data=sumtstv; 
 by group; 
run; 

 
%let extend=0.07; /*extend from middle to both side for the box position*/ 



• Create the annotate outlier dataset called "anno_out": In our customized box plot, any outliers outside the 
chosen bound will be shown as solid red dots.  

1. Get outlier data: 
data outlier; 
 merge tstv(in=a) sumtstv(keep=group lower upper); 
 by group; 
 if tstv>upper or tstv<lower then output; 
 keep group tstv; 
run; 
 

2.  Set annotate dataset "anno_out" for outliers: x=group; y=tstv.  
Function color style Text xsys ysys group tstv size position x y 
SYMBOL red NONE DOT 2 2 1 95.06078 3 5 1 95.06078 
SYMBOL red NONE DOT 2 2 .. .. 3 5 .. .. 
SYMBOL red NONE DOT 2 2 2 72.69134 3 5 2 72.69134 

 
• Build the annotate dataset "box" used for plotting the box plot and legend: The detail coding is provided in 

the appendix section. 
 
• Combine all annotate datasets together for box plot.  

data anno; 
 set box anno_out special; 
run; 
 

•  Define the plot area for using an annotate dataset as input to PROC GPLOT: Assign the y-axis scale by 
using the maximum of y-axis value and use dataset z to make plot area via gplot. 

   proc sql noprint; 
 select min(y), max(y) into :min, :max 
 from anno; 
quit; 

 
data z; 
 xz=7; yz=&max;  
run; 
 

• Create the customized boxplot by using annotate dataset and PROC GPLOT: 
axis1 label=none major=(h=.8)  value=(f=swissb h=2.5) offset=(5,1) 
   order=(1 to 6.3 by 1) minor=none 
   value=(t=1 h=3 j=c "10 mg" h=2.8 j=c "(N=&cnt1)" 
               t=2 h=3 j=c "20 mg" h=2.8 j=c "(N=&cnt2)" 
               t=3 h=3 j=c "30 mg" h=2.8 j=c "(N=&cnt3)" 
               t=4 h=3 j=c "100 mg" h=2.8 j=c "(N=&cnt4)" 
               t=5 h=3 j=c "200 mg" h=2.8 j=c "(N=&cnt5)"   
               t= 6 h=3 " "); 
axis2 label=(a=90 h=3 f=swissb "Endpoint") value=(f=swissb h=3) 
      minor=( 0.4 ber=9   h= num )
      order=(60 to 130 by 10) offset=(5, 5) major=(h=0.8); 
      title h=3.2 "Distribution by Treatment" ; 
proc gplot data=z ; 
 plot yz*xz / anno=anno 
                frame 
                haxis=axis1 
                vaxis=axis2 
                nolegend; 
run; quit; 



 

 
 
CONCLUSION 
The annotate facility is a widely used tool for generating a customized plot. By creating a well-designed 
annotate dataset, we can overcome much of the limitation imposed by certain SAS procedures and therefore 
achieve a high degree of customization. 

APPENDIX: 
Coding example for annotate dataset:box 
 



data box; 
   length function color style $ 8 text $ 20 position $ 1; 
   retain xsys '2' ysys '2' color ; 
   drop lower q1 median mean q3 upper; 
   set sumtstv; 
 
    line   =  0; 
    style  = 'S'; 
    color='lig'; 
   /** draw the box outline:extend from the middle point to draw the box**/ 
    function = 'MOVE'; x = group - &extend; y = q1; output; 
    function = 'BAR' ; x = group + &extend; y = q3; output; 
 
   /** draw the box board line **/ 
    line=1; 
    size=2; 

color='black'; 
 

    function = 'MOVE'; x = group-&extend; y = q3; output; 
    function = 'DRAW'; x = group+&extend; y = q3; output; 
    function = 'DRAW'; x = group+&extend; y = q1; output; 
    function = 'DRAW'; x = group-&extend; y = q1; output; 
    function = 'DRAW'; x = group-&extend; y = q3; output; 
    
   /** draw the upper whisker **/ 
    function = 'MOVE'; x = group;           y = q3;    output; 
    function = 'DRAW'; x = group;           y = upper; output; 
    function = 'MOVE'; x = group - &extend; y = upper; output; 
    function = 'DRAW'; x = group + &extend; y = upper; output; 
  
    /** draw the lower whisker **/ 
    function = 'MOVE'; x = group;           y = q1;    output; 
    function = 'DRAW'; x = group;           y = lower; output; 
    function = 'MOVE'; x = group - &extend; y = lower; output; 
    function = 'DRAW'; x = group + &extend; y = lower; output; 
  
    /** draw diamond for the median **/ 
    size     = 3; 
    position = '5';          /* position=5: center*/ 
    style    = 'NONE'; 
    function = 'SYMBOL';  text = 'DIAMOND'; 
    x = group;          
    y = median; 
    color='black'; output; 
 
    /** draw a dashed line for the mean **/ 
    line = 2; 
    size = 2; 
    function = 'MOVE';   x = group - &extend; y = mean; output; 
    function = 'DRAW';   x = group + &extend; y = mean; output; 
 
    /** the following is for the legend **/ 
    line   =  0; 
    style  = 'S'; 
    color='lig'; 
    left=5.2; middle=5.27; right=5.34; 
 



    /** draw the box outline **/ 
    function = 'MOVE'; x = left;   y = 119; output; 

function = 'BAR' ; x = right;  y = 111; output; 
 

    /** draw the box boarder line **/ 
    line=1; 
    size=2; 
    color='black'; 
    function = 'MOVE'; x = left;  y = 119; output; 
    function = 'DRAW'; x = right; y = 119; output; 
    function = 'DRAW'; x = right; y = 111; output; 
    function = 'DRAW'; x = left;  y = 111; output; 
    function = 'DRAW'; x = left;  y = 119; output; 

 
    /** draw the upper whisker **/ 
    function = 'MOVE'; x = middle; y = 119; output; 
    function = 'DRAW'; x = middle; y = 124; output; 
    function = 'MOVE'; x = left;   y = 124; output; 
    function = 'DRAW'; x = right;  y = 124; output; 
  
    /** draw the lower whisker **/ 
    function = 'MOVE'; x = middle; y = 111; output; 
    function = 'DRAW'; x = middle; y = 106; output; 
    function = 'MOVE'; x = left;   y = 106; output; 

function = 'DRAW'; x = right;  y = 106; output; 
 

    /** draw diamond for the median **/ 
    size     = 3; 
    position = '5'; 
    style    = 'NONE'; 
    function = 'SYMBOL';  text = 'DIAMOND'; 
    x = middle;          
    y = 116.5; 

color='black'; output; 
 

    /** draw a dashed line for the mean **/ 
    line = 2; 
    size = 2; 
    function = 'MOVE';   x = left;  y = 115; output; 
    function = 'DRAW';   x = right; y = 115; output; 
 
   /** draw the label **/     
    size=3; style='CENTB'; function='LABEL'; position='6'; x=left; y=128;  
    text='Legend'; output; 
    size=2;function='LABEL'; position='6'; x=right+0.1; y=124.5; 

text='Mean+SD'; output; 
    function='LABEL'; position='6'; x=right+0.1; y=119.7;  
    text='75th Percentile'; output; 
    function='LABEL'; position='6'; x=right+0.1; y=117;  
    text='Median'; output; 
    function='LABEL'; position='6'; x=right+0.1; y=115.5;  
    text='Mean'; output; 
    function='LABEL'; position='6'; x=right+0.1; y=111.2;  
    text='25th Percentile'; output; 
    function='LABEL'; position='6'; x=right+0.1; y=106.5;  

text='Mean-SD'; output; 
  run; 

